

DTMF Generation with a 3.58 MHz Crystal

National Semiconductor
Application Note 666
Verne H. Wilson
June 1990



DTMF Generation with a 3.58 MHz Crystal

DTMF (Dual Tone Multiple Frequency) is associated with digital telephony, and provides two selected output frequencies (one high band, one low band) for a duration of 100 ms. DTMF generation consists of selecting and combining two audio tone frequencies associated with the rows (low band frequency) and columns (high band frequency) of a push-button touch tone telephone keypad.

This application note outlines two different methods of DTMF generation using a COP820C/840C microcontroller clocked with a 3.58 MHz crystal in the divide by 10 mode. This yields an instruction cycle time of 2.79 μ s. The application note also provides a low true row/column decoder for the DTMF keyboard.

The first method of DTMF generation provides two PWM (Pulse Width Modulation) outputs on pins G3 and G2 of the G port for 100 ms. These two PWM outputs represent the selected high band and low band frequencies respectively, and must be combined externally with an LM324 op amp or equivalent feed back circuit to produce the DTMF signal.

The second method of DTMF generation uses ROM lookup tables to simulate the two selected DTMF frequencies. These table lookup values for the selected high band and low band frequencies are then combined arithmetically. The high band frequencies contain a higher bias value to compensate for the DTMF requirement that the high band frequency component be 2 dB above the low band frequency component to compensate for losses in transmission. The resultant value from the arithmetic combination of sine wave values is output on L port pins L0 to L5, and must be combined externally with a six input resistor ladder network to produce the DTMF signal. This resultant value is updated every 118 μ s. The COP820C/840C timer is used to time out the 100 ms duration of the DTMF. A timer interrupt at the end of the 100 ms is used to terminate the DTMF output. The external ladder network need not contain any active components, unlike the first method of DTMF generation with the two PWM outputs into the LM324 op amp.

The associated COP820C/840C program for the DTMF generation is organized as three subroutines. The first subroutine (KBRDEC) converts the low true column/row input from the DTMF keyboard into the associated DTMF hexadecimal digit. In turn, this hex digit provides the input for the other two subroutines (DTMFGP and DTMFLP), which represent the two different methods of DTMF generation. These three subroutines contain 35, 94, and 301 bytes of COP820C/840C code respectively, including all associated ROM tables. The Program Code/ROM table breakdowns are 19/16, 78/16, and 88/213 bytes respectively.

DTMF KEYBOARD MATRIX

The matrix for selecting the high and low band frequencies associated with each key is shown in *Figure 1*. Each key is uniquely referenced by selecting one of the four low band frequencies associated with the matrix rows, coupled with selecting one of the four high band frequencies associated with the matrix columns. The low band frequencies are

697 Hz, 770 Hz, 852 Hz, and 941 Hz, while the high band frequencies are 1209 Hz, 1336 Hz, 1477 Hz, and 1633 Hz. The DTMF keyboard input decode subroutine assumes that the keyboard is encoded in a low true row/column format, where the keyboard is strobed sequentially with four low true column selects with each returning a low true row select. The low true column and row selects are encoded in the upper and lower nibbles respectively of the accumulator, which serves as the input to the DTMF keyboard input decode subroutine. The subroutine will then generate the DTMF hexadecimal digit associated with the DTMF keyboard input digit.


The DTMF keyboard decode subroutine (KBRDEC) utilizes a common ROM table lookup for each of the two nibbles representing the low true column and row encodings for the keyboard. The only legal low true nibbles for a single key input are E, D, B, and 7. All other low true nibble values represent multiple keys, no key, or no column strobe. Results from two legal nibble table lookups (from the same 16 byte ROM table) are combined to form a hex digit with the binary format of 0000RRCC, where RR represents the four row values and CC represents the four column values. The illegal nibbles are trapped, and the subroutine is exited with a RET (return) command to indicate multiple keys or no key. A pair of legal nibble table lookups result in the subroutine being exited with a RETSK (return and skip) command to indicate a single key input. This KBRDEC subroutine uses 35 bytes of code, consisting of 19 bytes of program code and 16 bytes of ROM table.

DTMF GENERATION USING PWM AND AN OP AMP

The first DTMF generation method (using the DTMFGP subroutine) generates the selected high band and low band frequencies as PWM (Pulse Width Modulation) outputs on pins G3 and G2 respectively of the G port. The COP820C/840C microcontrollers each contain only one timer, and three times must be generated to satisfy the DTMF application. These three times are the half periods of the two selected frequencies and the 100 ms duration period. Obviously the single timer can only generate one of the required times, while the program must generate the two remaining times. The solution lies in dividing the 100 ms duration time by the half periods for each of the eight DTMF frequencies, and then examining the respective high band and low band quotients and remainders. Naturally these divisions must be normalized to the instruction cycle time (t_C). 100 ms represents 35796 t_C 's. The results of these divisions are detailed in Table I.

The four high band frequencies are produced by running the COP820C/840C timer in PWM (Pulse Width Modulation) mode, while the program produces the four low band frequencies and the 100 ms duration timeout. The programmed times are achieved by using three programmed register counters R0, R2 and R3, with a backup register R1 to reload the counter R0. These three counters represent the half period, the 100 ms quotient, and the 100 ms remainder associated with each of the four low band frequencies.

AN-666



TL/DD/10740-22

FIGURE 1. DTMF Keyboard Matrix

TABLE I. Frequency Half Periods, Quotients and Remainders

	Freq. Hz	Half Period in μ s	Half Period in t_c 's	100 ms/0.5P in t_c 's	
				Quotient	Remainder
Low Band Frequencies	697	717.36	257	139	73
	770	649.35	232	154	68
	852	586.85	210	170	96
	941	531.35	190	188	76
High Band Frequencies	1209	413.56	148	241	128
	1336	374.25	134	267	18
	1477	338.53	121	295	101
	1633	306.18	110	325	46

Note: 100 ms represents 35796 t_c 's.

The theory of operation in producing the selected low band frequency starts with loading the three counters with values obtained from a ROM table. The half period for the selected frequency is counted out, after which the G2 output bit is toggled. During this half period countout, the quotient counter is decremented. This procedure is repeated until the quotient counter counts out, after which the program branches to the remainder loop. During the remainder loop, the remainder counter counts out to terminate the 100 ms. Following the remainder countout, the G2 and G3 bits are both reset, after which the DTMF subroutine is exited. Great care must be taken in time balancing the half period loop for

the selected low band frequency. Furthermore, the toggling of the G2 output bit (achieved with either a set or reset bit instruction) must also be exactly time balanced to maintain the half period time integrity. Local stall loops (consisting of a DRSZ instruction followed by a JP jump back to the DRSZ for a two byte, six instruction cycle loop) are embedded in both the half period and remainder loops. Consequently, the ROM table parameters for the half period and remainder counters are approximately only one-sixth of what otherwise might be expected. The program for the half period loop, along with the detailed time balancing of the loop for each of the low band frequencies, is shown in *Figure 2*.

	Program	Bytes/ Cycles	Conditional Cycles	Cycles	Total Cycles
	LD B,#PORTGD	2/3			
	LD X,#R1	2/3			
LUP1:	LD A,[X-]	1/3		3	
	IFBIT 2,[B]	1/1		1	
	JP BYP1	1/3	3	1	
	X A,[X+]	1/3		3	
	SBIT 2,[B]	1/1		1	
	JP BYP2	1/3		3	
BYP1:	NOP	1/1	1		
	RBIT 2,[B]	1/1	1		
	X A,[X+]	1/3	3		
BYP2:	DRSZ R2	1/3		3	
	JP LUP2	1/3		3	
	JP FINI	1/3			
LUP2:	DRSZ R0	1/3	3	3	
	JP LUP2	1/3	3	1	
	LD A,[X]	1/3		3	
	IFEQ A,#31	2/2		2	
	JP LUP1	1/3	1	3	30
	NOP	1/1	1		
	NOP	1/1	1		
	IFEQ A,#38	2/2	2		
	JP LUP1	1/3	1	3	35
	LAIID	1/3	3		
	NOP	1/1	1		
	JP LUP1	1/3	3		40

Table III Frequency	Stall Loop	Total Cycles	Half Period
[(38 - 1)	× 6]	+ 35	= 257
[(33 - 1)	× 6]	+ 40	= 232
[(31 - 1)	× 6]	+ 30	= 210
[(26 - 1)	× 6]	+ 40	= 190

FIGURE 2. Time Balancing for Half Period Loop

TABLE IV. Time Balancing for Remainder Loop

Table III Remainder	Stall Loop	R Loop Overhead	Total Cycles	Table I Remainder
[(10 - 1)	× 6]	+ 20	= 74	73
[(9 - 1)	× 6]	+ 20	= 68	68
[(14 - 1)	× 6]	+ 20	= 98	96
[(10 - 1)	× 6]	+ 20	= 74	76

Note that the Q value in Table III is one greater than the quotient in Table I to compensate for the fact that the quotient count down to zero test is performed early in the half period loop. The overhead in the remainder loop is 20 instruction cycles. The detailed time balancing for the remainder loop is shown in Table IV.

The selected high band frequency is achieved by loading the half period count in t_C 's minus one (from Table III) into the timer autoreload register and running the timer in PWM output mode. The minus one is necessary since the timer toggles the G3 output bit when it underflows (counts down through zero), at which time the contents of the autoreload register are transferred into the timer.

In summary, the input digit from the keyboard (encoded in low true column/row format) is translated into a digit matrix vector XXXRRRCC which is checked for 1001RRCC to indicate a single key entry. No key or multiple key entries will set a flag and terminate the DTMF subroutine. The digit matrix vector for a single key is transformed into the core vector 0000RRCC. The core vector is then translated into four other vectors (T, F, Q, R) which in turn are used to select four parameters from a 16 byte ROM table. These four parameters are used to load the timer, and the respective half period, quotient, and remainder counters. The 16 byte ROM table must be located starting at ROM location 0030 (or 0X30) in order to minimize program size, and has reference setups with the "OR A, #033" instruction for the F vector and the "OR A, #030" instruction for the T vector.

The three parameters associated with the two R bits of the core vector require a multi-level table lookup capability with the LAID instruction. This is achieved with the following section of code in the DTMF subroutine:

```


LD      B, #R1
LUP:   X      A, [B]
LD      A, [B,]
LAID
X      A, [B+]
DEC     A
IFBNE  #4
JP      LUP
    
```

This program loads the F frequency vector into R1, and then decrements the vector each time around the loop. The vector is successively moved with the exchange commands from R1 to R2 to R3 as one of the same exchange commands loads the data from the ROM table into R1, R2, and R3. This successive decrementation of the F vector changes the F vector into the Q vector, and then changes the Q vector into the R vector. These vectors are used to access the ROM table with the LAID instruction. The B pointer is incremented each time around the loop after it has been used to store away the three selected ROM table parameters (one per loop). These three parameters are stored in sequential RAM locations R1, R2, and R3. The IFBNE test instruction is used to skip out of the loop once the three selected ROM table parameters have been accessed and stored away.

The timer is initialized to a count of 15 so that the first timer underflow and toggling of the G3 output bit (with timer PWM mode and G3 toggle output selected) will occur at the same time as the first toggling of the G2 output bit. The half period counts for the high band frequencies minus one are stored in the timer section of the ROM table. The selected value from this frequency ROM table is stored in the timer autoreload register. The timer is selected for PWM output mode and started with the instruction LD [B], #0B0 where the B pointer is selecting the CNTRL register at memory location 0EE.

This first DTMF generation subroutine for the COP820C/840C uses 94 bytes of code, consisting of 78 bytes of program code and 16 bytes of ROM table. A program test routine to sequentially call the DTMFGP subroutine for each of the 16 keyboard input digits is supplied with the listing for the DTMF35 program. This test routine uses a 16 byte ROM table to supply the low true encoded column/row keyboard input to the accumulator. An input from the I0 input pin of the I port is used to select which DTMF generation subroutine is to be used. The DTMFGP subroutine is selected with I0 = 0.

A TYPICAL OP AMP CONFIGURATION FOR MIXING THE TWO DTMF PWM OUTPUTS IS SHOWN IN FIGURE 3.



TL/DD/10740-23

FIGURE 3. Typical Op Amp Configuration for Mixing DTMF PWM Outputs

DTMF GENERATION USING A RESISTOR LADDER NETWORK

The second DTMF generation method (using the DTMF sub-routine) generates and combines values from two table lookups simulating the two selected sine waves. The high band frequency table values have a higher base line value (16 versus 13) than the low band frequency table values. This higher bias for the high frequency values is necessary to satisfy the DTMF requirement that the high band DTMF frequencies need a value 2 dB greater than the low band DTMF frequencies to compensate for losses in transmission.

The resultant value from arithmetically combining the table lookup low band and high band frequency values is output on pins L0 to L5 of the L port in order to feed into a six input external resistor ladder network. The resultant value is updated every $117\frac{1}{3} \mu\text{s}$ (one cycle of the LUP42 program loop). The LUP42 program loop contains 42 instruction cycles (t_C 's) of $2.7936511 \mu\text{s}$ each for a total loop time of $117\frac{1}{3} \mu\text{s}$. The COP820C/840C timer is used to count out the 100 ms DTMF duration time.

An interrupt from the timer terminates the 100 ms DTMF output. Note that the Stack Pointer (SP) must be adjusted following the timer interrupt before returning from the DTMF sub-routine.

The DTMF sub-routine starts by quadrupling the value of the DTMF hex digit value in the accumulator, and then adding an offset value to reach the first value in the telephone key table. The telephone key ROM table contains four values associated with each of the 16 DTMF hex keys. These four values represent the low and high frequency table sizes and table starting addresses associated with the pair of frequencies (one low band, one high band) associated with each DTMF key. The FRLUP section of the program loads the four associated telephone key table values from the ROM table into the registers LFTBSZ (Low Freq Table Size), LFTADR (Low Freq Table Address), HFTBSZ (High Freq Table Size), and HFTADR (High Freq Table Address). The program then initializes the timer and autoreload register, starts the timer, and then jumps to LUP42. Note that the timer value in t_C 's is 100 ms plus one LUP42 time, since the initial DTMF output is not until the end of the LUP42 program.

Multiples of the magic number $118 \mu\text{s}$ (approximately) are close approximations to all eight of the DTMF frequencies. The LUP42 program uses 42 instruction cycles (of $2.7936511 \mu\text{s}$ each) to yield a LUP42 time of $117\frac{1}{3} \mu\text{s}$. The purpose of the LUP42 program is to update the six L port outputs by accessing and then combining the next set of

values from the selected low band and high band sine wave frequency tables in the ROM. The ROM table offset frequency pointers (LFPTR and HFPTR) must increment each time and then wrap around from top to bottom of the two selected ROM tables. The ROM table size parameters (LFTBSZ and HFTBSZ) for the selected frequencies are tested during each LUP42 to determine if the wrap around from ROM table top to bottom is necessary. The wrap around is implemented by clearing the frequency pointer in question. Note that the ROM tables are mapped from a reference of 0 to table size minus one, so that the table size is used in a direct comparison with the frequency offset pointer to test for the need for a wrap around. Also note that the offset pointer incremented value is used during the following LUP42 cycle, while the pre-incremented value of the pointer is used during the current cycle. However, it is the incremented value that is tested versus the table size for the need to wrap around.

After the low band and high band ROM table sine wave frequency values are accessed in each cycle of the LUP42 program, they are added together and then output to pins L0–L5 of the L port. As stated previously, the low band frequency values have a lower bias than the high band frequency values to compensate for the required 2 dB offset. Specifically, the base line and maximum values for the low frequency values are 13 and 26 respectively, while the base line and maximum values for the high frequency values are 16 and 32 respectively. Thus the combined base line value is 29, while the combined maximum value is 58. This gives a range of values on the L port output (L0–L5) from 0 to 58.

The minimum time necessary for the LUP42 update program loop is 36 instruction cycles including the jump back to the start of the loop. Consequently, two LAID instructions are inserted just prior to the jump back instruction at the end of LUP42 to supply the six extra NOP instruction cycles needed to increase the LUP42 instruction cycles from 36 to 42. A three cycle LAID instruction can always be used to simulate three single cycle NOP instructions if the accumulator data is not needed.

Table V shows the multiple LUP42 approximation to the eight DTMF frequencies, including the number of sine wave cycles and data points in the approximation. As an example, three cycles of a sine wave with a total of 19 data points across the three cycles is used to approximate the 1336 Hz DTMF frequency. The 19 cycles of LUP42 times the LUP42 time of $117\frac{1}{3} \mu\text{s}$ is divided into the three cycles to yield a value of 1345.69 Hz. This gives an error of +0.73% when compared with the DTMF value of 1336 Hz. This is well within the 1.5% North American DTMF error range.

TABLE V. DTMF Frequency Approximation Table

DTMF Freq.	# of Sine Wave Cycles	# of Data Points	Calculation	Approx. Freq.	% Error
697	4	49	$4/(49 \times 117\frac{1}{3})$	= 695.73	-0.18
770	1	11	$1/(11 \times 117\frac{1}{3})$	= 774.79	+0.62
852	1	10	$1/(10 \times 117\frac{1}{3})$	= 852.27	+0.03
941	1	9	$1/(9 \times 117\frac{1}{3})$	= 946.97	+0.63
1209	1	7	$1/(7 \times 117\frac{1}{3})$	= 1217.53	+0.71
1336	3	19	$3/(19 \times 117\frac{1}{3})$	= 1345.69	+0.73
1477	4	23	$4/(23 \times 117\frac{1}{3})$	= 1482.21	+0.35
1633	4	21	$4/(21 \times 117\frac{1}{3})$	= 1623.38	-0.59

The frequency approximation is equal to the number of cycles of sine wave divided by the time in the total number of LUP42 cycles before the ROM table repeats.

The values in the DTMF sine wave ROM tables are calculated by computing the sine value at the appropriate points, scaling the sine value up to the base line value, and then adding the result to the base line value. The following example will help to clarify this calculation.

Consider the three cycles of sine wave across 19 data points for the 1336 Hz high band frequency. The first value in the table is the base line value of 16. With 2π radians per sine wave cycle, the succeeding values in the table represent the sine values of $1 \times (6\pi/19)$, $2 \times (6\pi/19)$, $3 \times (6\pi/19)$, . . . , up to $18 \times (6\pi/19)$. Consider the seventh and eighth values in the table, representing the sine values of $6 \times (6\pi/19)$ and $7 \times (6\pi/19)$ respectively. The respective calculations of $16 \times \sin[6 \times (6\pi/19)]$ and $16 \times \sin[7 \times (6\pi/19)]$ yield values of -5.20 and 9.83 . Rounding to the nearest integer gives values of -5 and 10 . When added to the base line value of 16, these values yield the results 11 and 26 for the seventh and eighth values in the 1336 Hz DTMF ROM table. Symmetry in the loop of 19 values in the DTMF table dictates that the fourteenth and thirteenth values in the table are 21 and 6, representing values of 5 and -10 from the calculations.

The area under a half cycle of sine wave relative to the area of the surrounding rectangle is $2/\pi$, where π radians represent the sine wave half cycle. This surrounding rectangle has a length of π and a height of 1, with the height representing the maximum sine value. Consequently, the area of the surrounding rectangle is π . The integral of the area under the half sine wave from 0 to π is equal to 2. The ratio of $2/\pi$ is equal to 63.66%, so that the total of the values for each half sine wave should approximate 63.66% of the sum of the max values. The maximum values (relative to the base line) are 13 and 16 respectively for the low and high band DTMF frequencies.

For the previous 1336 Hz example, the total of the absolute values for the 19 sine values from the 1336 Hz ROM

table is equal to 196. The surrounding rectangle for the three cycles of sine wave is 19 by 16 for a total area of 304. The ratio of $196/304$ is 64.47% compared with the $2/\pi$ ratio of 63.66%. Thus the sine wave approximation gives an area abundance of 0.81% (equal to $64.47 - 63.66$).

An application of the sine wave area criteria is shown in the generation of the DTMF 852 Hz frequency. The ten sine values calculated are 0, 7.64, 12.36, 12.36, 7.64, 0, -7.64 , -12.36 , -12.36 , and -7.64 . Rounding off to the nearest integer yields values of 0, 8, 12, 12, 8, 0, -8 , -12 , -12 and -8 . The total of these values (absolute numbers) is 80, while the area of the surrounding rectangle is 130 (10×13). The ratio of $80/130$ is 61.54% compared with the $2/\pi$ ratio of 63.66%. Thus the sine wave approximation gives an area deficiency of 2.12% (equal to $63.66 - 61.54$), which is overly deficient. Consequently, two of the ten sine values are augmented to yield sine values of 0, 8, 12, 13*, 8, 0, -8 , -12 , -13^* , and -8 . This gives an absolute total of 82 and a ratio of $82/130$, which equals 63.08% and serves as a much better approximation to the $2/\pi$ ratio of 63.66%.

The sine wave area criteria is also used to modify two values in the DTMF 941 Hz frequency. The nine sine values calculated are 0, 8.36, 12.80, 11.26, 4.45, -4.45 , -11.26 , -12.80 , and -8.36 . Rounding off to the nearest integer yields values of 0, 8, 13, 11, 4, -4 , -11 , -13 , and -8 . The total of these values (absolute numbers) is 72, while the area of the surrounding rectangle is 117 (9×13). The ratio of $72/117$ is 61.54% compared to the $2/\pi$ ratio of 63.66%. Thus the sine wave approximation gives an area deficiency of 2.12% (equal to $63.66 - 61.54$), which is overly deficient. Rounding up the two values of 4.45 and -4.45 to 5 and -5 , rather than down to 4 and -4 , yields values of 0, 8, 13, 11, 5, -5 , -11 , -13 and -8 . This gives an absolute total of 74 and a ratio of $74/117$, which equals 63.25% and serves as a much better approximation to the $2/\pi$ ratio of 63.66%.

With these modified values for the 852 and 941 DTMF frequencies, the area criteria ratio of $2/\pi = 63.66\%$ for the sine wave compared to the surrounding rectangle has the following values:

DTMF Freq.	Sum of Values	Rectangle Area	Percentage	Diff.
697 Hz	406	$49 \times 13 = 637$	63.74%	+0.08%
770 Hz	92	$11 \times 13 = 143$	64.34%	+0.68%
852 Hz	82	$10 \times 13 = 130$	63.08%	-0.58%
941 Hz	74	$9 \times 13 = 117$	63.25%	-0.41%
1209 Hz	72	$7 \times 16 = 112$	64.29%	+0.63%
1336 Hz	196	$19 \times 16 = 304$	64.47%	+0.81%
1477 Hz	232	$23 \times 16 = 368$	63.04%	-0.62%
1633 Hz	216	$21 \times 16 = 336$	64.29%	+0.63%

The LUP42 program loop is interrupted by the COP820C/840C timer after 100 ms of DTMF output. As stated previously, the Stack Pointer (SP) must be adjusted (incremented by 2) following the timer interrupt before returning from the DTMFLP subroutine.

This second DTMF generation subroutine for the COP820C/840C uses 301 bytes of code, consisting of 88 bytes of program code and 213 bytes of ROM table. The following is a summary of the DTMFLP subroutine code allocation.

DTMFLP Code Allocation	# of Bytes
1. Subroutine Header Code	42
2. Interrupt Code	16
3. LUP42 Code	30
4. Telephone Key Table	64
5. Sine Value Tables	149
Total	301

A program test routine to sequentially call the DTMFLP subroutine for each of the 16 DTMF keyboard input digits is supplied with the listing for the DTMF35 program. This test routine uses a 16 byte ROM table to supply the low true encoded column/row keyboard input to the accumulator. An input from the I0 pin of the I port is used to select which DTMF generation subroutine is to be used. The DTMFLP subroutine is selected with I0 = 1.

A TYPICAL RESISTOR LADDER NETWORK IS SHOWN IN FIGURE 4.

SUMMARY

In summary, the DTMF35 program assumes a COP820C/840C clocked with a 3.58 MHz crystal in divide by 10 mode. The DTMF35 program contains three subroutines, KBRDEC, DTMFGP, and DTMFLP. The KBRDEC subroutine is a low true DTMF keyboard decoder, while the DTMFGP and DTMFLP subroutines represent the alternative methods of DTMF generation.

The KBRDEC subroutine provides a low true decoding of the DTMF keyboard input and assumes that the keyboard input has been encoded in a low true column/row format, with the columns of the keyboard being sequentially strobed.

The DTMFGP subroutine produces two PWM (Pulse Width Modulation) outputs (representing the selected high and low band DTMF frequencies) for combination with an external op amp network (LM324 or equivalent).

The DTMFLP subroutine produces six bits of combined high band and low band DTMF frequency output for combination in an external resistor ladder network. This output represents a combined sine wave simulation of the two selected DTMF frequencies by combining values from two selected ROM tables, and updating these values every 118 μ s.

The three DTMF35 subroutines contain the following number of bytes of program and ROM table memory:

Subroutine	# of Bytes of Program	# of Bytes of ROM Table	Total # of Bytes
KBRDEC	19	16	35
DTMFGP	78	16	94
DTMFLP	88	213	301




FIGURE 4. Typical Resistor Ladder Network

TL/DD/10740-24


```

1      ;
2      ; DTMF GENERATION WITH A 3.58 MHZ      VERNE H. WILSON
3      ;           CRYSTAL FOR COP820C/840C      10/28/89
4      ;
5      ; DTMF - DUAL TONE MULTIPLE FREQUENCY
6      ;
7      ; PROGRAM NAME: DTMF35.MAC
8      ;
9      ;           .TITLE DTMF35
10     ;           .CHIP 840
11     ;
12     ;
13     ; THIS DTMF PROGRAM IS BASED ON A COP820C/840C RUNNING
14     ; WITH A CKI CLOCK OF 3.579545 MHZ (TV COLOR CRYSTAL
15     ; FREQUENCY) IN DIVIDE BY 10 MODE, FOR AN INSTRUCTION
16     ; CYCLE TIME OF 2.7936511 MICROSECONDS.
17     ;
18     ; THIS PROGRAM CONTAINS THREE SUBROUTINES, ONE FOR A
19     ; LOW TRUE ROW/COLUMN DTMF KEYBOARD DECODING (KBRDEC),
20     ; AND THE OTHER TWO (DTMFGP, DTMFLP) FOR ALTERNATE
21     ; METHODS OF DTMF GENERATION.
22     ;
23     ; KEYBOARD INPUT DATA IS IN ACCUMULATOR WITH A
24     ; LOW TRUE FORMAT AS FOLLOWS:
25     ;     BITS 7 TO 4 : LOW TRUE COLUMN VALUE (E,D,B,7)
26     ;     BITS 3 TO 0 : LOW TRUE ROW VALUE (E,D,B,7)
27     ;
28     ; ASSUMPTION MADE THAT COLUMN STROBES (LOW TRUE) ARE
29     ; OUTPUT, WHILE ROW VALUES (LOW TRUE) ARE INPUT.
30     ;
31     ; THE FIRST METHOD OF DTMF GENERATION CONSISTS OF
32     ; GENERATING TWO PWM OUTPUTS ON THE G PORT G2 AND G3
33     ; OUTPUT PINS. THESE TWO OUTPUTS NEED TO BE MIXED
34     ; EXTERNALLY WITH AN APPROPRIATE LM324 OP AMP FEEDBACK
35     ; CIRCUIT TO GENERATE THE DTMF.
36     ;
37     ; THE SECOND METHOD OF DTMF GENERATION USES ROM LOOKUP
38     ; TABLES TO SIMULATE THE TWO DTMF SINE WAVES AND
39     ; COMBINES THEM ARITHMETICALLY. THE RESULT IS OUTPUT ON
40     ; THE LOWER SIX BITS OF THE L PORT (L0 - L5). THESE SIX
41     ; OUTPUTS ARE COMBINED EXTERNALLY WITH A LADDER NETWORK
42     ; TO GENERATE THE DTMF.
43     ;
44     ; THE SECOND DTMF GENERATION METHOD USES APPROXIMATELY
45     ; THREE TIMES AS MUCH ROM CODE (INCLUDING PROGRAM CODE
46     ; AND ROM TABLES) AS THE FIRST METHOD, BUT HAS THE
47     ; ADVANTAGE OF ELIMINATING THE COST OF THE EXTERNAL
48     ; ACTIVE COMPONENT (LM324 OR EQUIVALENT).
49     ;
50     ; BOTH OF THE DTMF SUBROUTINES GENERATE THEIR OUTPUTS
51     ; FOR A PERIOD OF 100 MILLISECONDS.

```

TL/DD/10740-1

```

52      ;
53      ; DECLARATIONS:
54      ;
55      0000      KDATA = 0      ; *** KEYBOARD DATA ***
56      00D0      PORTLD = 0D0    ; PORTL DATA REG
57      00D1      PORTLC = 0D1    ; PORTL CONFIG REG
58      00D4      PORTGD = 0D4    ; PORTG DATA REG
59      00D5      PORTGC = 0D5    ; PORTG CONFIG REG
60      00D7      PORTI = 0D7     ; PORTI INPUT PINS
61      00DC      PORTD = 0DC     ; PORTD REG
62      00EA      TMRLO = 0EA     ; TIMER LOW COUNTER
63      00EB      TMRHI = 0EB     ; TIMER HIGH COUNTER
64      00EC      TAULO = 0EC     ; TMR AUTORELOAD REG LO
65      00ED      TAUHI = 0ED     ; TMR AUTORELOAD REG HI
66      00EE      CNTRL = 0EE     ; CONTROL REG
67      00EF      PSW = 0EF       ; PROC STATUS WORD
68      00F0      R0 = 0F0        ; LB FREQ LOOP COUNTER
69      00F1      R1 = 0F1        ; LB FREQ LOOP COUNT
70      00F2      R2 = 0F2        ; LB FREQ Q COUNT
71      00F3      R3 = 0F3        ; LB FREQ R COUNT
72      ;
73      0000 DD2F      START: LD      SP,#02F      ; INITIALIZE STACK PTR
74      ;
75      ;
76      ;
77      0002 DEDC      LD      B,#PORTD      ;
78      0004 9E00      LD      [B],#0      ;
79      0006 A0        LOOP: RC      ; * 0 # D
80      0007 AE        LD      A,[B]      ; DTMF TEST LOOP
81      0008 9405      ADD     A,#5      ; SEQUENCE IS 1,5,9,D,4,
82      000A A6        X      A,[B]      ; 8,#,A,7,0,3,B,*,2,6,C
83      000B 6C        RBIT   4,[B]      ; HEX MATRIX TO LOOKUP
84      000C 9420      ADD     A,#020     ; TABLE FOR LOW TRUE
85      000E A4        LAID   ; COLUMN/ROW INPUT TO
86      000F 3210      JSR    KBRDEC     ; KBRDEC SUBROUTINE
87      0011 A1        SC      ; SET C IF NOT SINGLE KEY
88      0012 DED7      LD      B,#PORTI   ; TEST BIT 0 OF PORTI TO
89      0014 70        IFBIT 0,[B]      ; DETERMINE WHICH
90      0015 03        JP      BYPA      ; DTMF SUBROUTINE
91      0016 3040      JSR    DTMFGP     ; TWO PWM OUTPUTS ON
92      0018 02        JP      BYPB      ; G PORT PINS G2,G3
93      0019 308E      BYPA: JSR    DTMFLP ; SIX LADDER OUTPUTS ON
94      ;
95      001B DEDC      BYPB: LD      B,#PORTD ; DO WILL TOGGLE FOR EACH
96      001D E8        JP      LOOP      ; CALL OF SUBROUTINE
97      ;
98      ;
99      ;

```

TL/DD/10740-2

```

100                                     .FORM
101                                     ;
102                                     ; KEYBOARD DIGIT MATRIX TABLE
103                                     ;
104         0020                         . = 020
105                                     ;
106                                     ;
107 0020 EE                             .BYTE      1  5  9  D  4  8  #  A
        0021 DD
        0022 BB
        0023 77
        0024 ED
        0025 DB
        0026 B7
        0027 7E
108                                     ;
109 0028 EB                             .BYTE      7  0  3  B  *  2  6  C
        0029 D7
        002A BE
        002B 7D
        002C E7
        002D DE
        002E BD
        002F 7B
110                                     ;
111                                     ;
112                                     ;
113                                     ;
114                                     ;
115                                     ; FIRST DTMF SUBROUTINE (DTMFGP) PRODUCES TWO PWM
116                                     ; (PULSE WIDTH MODULATION) OUTPUTS ON PINS G3, G2
117                                     ;
118                                     ;
119                                     ; G PORT IS USED FOR THE TWO OUTPUTS
120                                     ; - HIGH BAND (HB) FREQUENCY OUTPUT ON G3
121                                     ; - LOW BAND (LB) FREQUENCY OUTPUT ON G2
122                                     ;
123                                     ; TIMER COUNTS OUT
124                                     ; - HB FREQUENCIES
125                                     ;
126                                     ; PROGRAM COUNTS OUT
127                                     ; - LB FREQUENCIES
128                                     ; - 100 MSEC DIVIDED BY LB HALF PERIOD QUOTIENT
129                                     ; - 100 MSEC DIVIDED BY LB HALF PERIOD REMAINDER
130                                     ;
131                                     ; NOTE THAT ALL COUNTS MUST BE NORMALIZED TO THE
132                                     ; 2.7936511 MICROSECOND INSTRUCTION CYCLE Tc
133                                     ;
134                                     ; 100 MSEC REPRESENTS 35796 Tc's
135                                     ;
136                                     ;

```

TL/DD/10740-3

```

137 ;
138 ;
139 ; HALF PERIODS FOR THE 8 DTMF FREQUENCIES (697,770,852,
140 ; 941,1209,1336,1477, AND 1633 KHZ) ARE 257,232,
141 ; 210,190,148,134,121, AND 110 Tc's RESPECTIVELY
142 ;
143 ; THE 100 MSEC DIVIDED BY HALF PERIOD QUOTIENTS ARE
144 ; 139,154,170,188,241,267,295, AND 325 RESPECTIVELY
145 ;
146 ; THE 100 MSEC DIVIDED BY HALF PERIOD REMAINDERS ARE
147 ; 72,67,95,75,127,17,100, AND 45 RESPECTIVELY
148 ;
149 ;
150 ;
151 ;
152 ; BINARY FORMAT FOR THE HEX DIGIT KEY VALUE FROM THE
153 ; KBRDEC SUBROUTINE IS 0000RRCC,
154 ; WHERE - RR IS ROW SELECT (LB FREQUENCIES)
155 ; - CC IS COLUMN SELECT (HB FREQUENCIES)
156 ;
157 ; FREQUENCY VECTORS (HB & LB) FOR FREQ PARAMETER TABLE
158 ; MADE FROM KEY VALUE
159 ;
160 ; HB FREQ VECTORS (4) END WITH 00 FOR TIMER COUNTS,
161 ; WHERE VECTOR FORMAT IS 0011CC00
162 ;
163 ; LB FREQUENCY VECTORS (12) END WITH:
164 ; 11 FOR HALF PERIOD LOOP COUNTS,
165 ; WHERE VECTOR FORMAT IS 0011RR11
166 ; 10 FOR 100 MSEC DIVIDED BY HALF PERIOD QUOTIENTS,
167 ; WHERE VECTOR FORMAT IS 0011RR10
168 ; 01 FOR 100 MSEC DIVIDED BY HALF PERIOD REMAINDERS,
169 ; WHERE VECTOR FORMAT IS 0011RR01
170 ;
171 ; FREQ PARAMETER TABLE AT HEX 003* (REQUIRED LOCATION)
172 ;
173 ;
174 ;
175 ; KEY VALUE
176 ; 0000RRCC
177 ;
178 ; TIMER T CCOO
179 ; R1 F RR11
180 ; R2 Q RR10
181 ; R3 R RR01
182 ;
183 ;
184 ;

```

TL/DD/10740-4

```

185                                     .FORM
186                                     ;
187                                     ;
188                                     ;FREQUENCY AND 100 MSEC PARAMETER TABLE
189                                     ;
190 0030 93                             .BYTE      147          ; T
191 0031 0A                             .BYTE      10           ; R
192 0032 8C                             .BYTE      140          ; Q
193 0033 26                             .BYTE      38           ; F
194 0034 85                             .BYTE      133          ; T
195 0035 09                             .BYTE      9           ; R
196 0036 9B                             .BYTE      155          ; Q
197 0037 21                             .BYTE      33           ; F
198 0038 78                             .BYTE      120          ; T
199 0039 0E                             .BYTE      14           ; R
200 003A AB                             .BYTE      171          ; Q
201 003B 1F                             .BYTE      31           ; F
202 003C 6D                             .BYTE      109          ; T
203 003D 0A                             .BYTE      10           ; R
204 003E BD                             .BYTE      189          ; Q
205 003F 1A                             .BYTE      26           ; F
206                                     ;
207                                     ;
208                                     ;
209 0040 DED5                           DTMFGP: LD      B,#PORTGC ; CONFIGURE G PORT
210 0042 9B3F                           LD      [B-],#03F ; FOR OUTPUTS
211 0044 6B                             RBIT    3,[B] ; OPTIONAL HB RESET
212 0045 6A                             RBIT    2,[B] ; OPTIONAL LB RESET
213 0046 5F                             LD      B,#KDATA
214 0047 A6                             X       A,[B] ; STORE KEY VALUE
215 0048 AE                             LD      A,[B] ; KEY VALUE TO ACC
216 0049 9733                           OR      A,#033 ; CREATE LB FREQ VECTOR
217 004B DEF1                           LD      B,#R1 ; FROM KEY VALUE
218 004D A6                             LUP:    X       A,[B]
219 004E AE                             LD      A,[B] ; THREE PARAMETERS
220 004F A4                             LAID    ; FROM LOW BAND
221 0050 A2                             X       A,[B+] ; FREQ ROM TABLE
222 0051 8B                             DEC     A ; TO R1,R2,R3
223 0052 44                             IFBNE   #4
224 0053 F9                             JP      LUP
225 0054 5F                             LD      B,#KDATA
226 0055 AE                             LD      A,[B] ; KEY VALUE TO ACC
227 0056 65                             SWAP    A ; CREATE HB FREQ VECTOR
228 0057 A0                             RC      ; FROM KEY VALUE
229 0058 B0                             RRC     A
230 0059 B0                             RRC     A
231 005A 9730                           OR      A,#030
232 005C A4                             LAID    ; HB FREQ TABLE
233 005D DEEA                           LD      B,#TMRLO ; (1 PARAMETER)
234 005F 9A0F                           LD      [B+],#15 ; INSTRUCTION CYCLE
235 0061 9A00                           LD      [B+],#0 ; TIME UNTIL TOGGLE

```

TL/DD/10740-5

```

236 0063 A2          X          A,[B+]      ; HB FREQ PARAMETER TO
237 0064 9A00        LD          [B+],#0    ; AUTORELOAD REGISTER
238 0066 9EB0        LD          [B],#0B0    ; START TIMER PWM
239 0068 DED4        LD          B,#PORTGD
240 006A DCF1        LD          X,#R1
241 006C BB          LUP1:      LD          A,[X-]
242 006D 72          IFBIT      2,[B]      ; TEST LB OUTPUT
243 006E 03          JP          BYP1
244 006F B2          X          A,[X+]
245 0070 7A          SBIT      2,[B]      ; SET LB OUTPUT
246 0071 03          JP          BYP2
247 0072 B8          BYP1:      NOP
248 0073 6A          RBIT      2,[B]      ; RESET LB OUTPUT
249 0074 B2          X          A,[X+]
250 0075 C2          BYP2:      DRSZ      R2          ; DECR. QUOT. COUNT
251 0076 01          JP          LUP2
252 0077 0E          JP          FINI      ; Q COUNT FINISHED
253 0078 C0          LUP2:      DRSZ      R0          ; DECR. F COUNT
254 0079 FE          JP          LUP2      ; LB (HALF PERIOD)
255 ;
256 007A BE          LD          A,[X]      ; *****
257 007B 921F        IFEQ      A,#31      ; BALANCE          ***
258 007D EE          JP          LUP1      ; LOW BAND         ***
259 007E B8          NOP          ; FREQUENCY        ***
260 007F B8          NOP          ; RESIDUE          ***
261 0080 9226        IFEQ      A,#38      ; DELAY FOR        ***
262 0082 E9          JP          LUP1      ; EACH OF THE      ***
263 0083 A4          LAID          ; FOUR LOW BAND    ***
264 0084 B8          NOP          ; FREQUENCIES      ***
265 0085 E6          JP          LUP1      ; *****
266 0086 C3          FINI:      DRSZ      R3          ; DECR. REMAINDER COUNT
267 0087 FE          JP          FINI      ; REM. COUNT NOT FINISHED
268 0088 BDEE6C      RBIT      4,CNTRL    ; STOP TIMER
269 008B 6B          RBIT      3,[B]      ; OPTIONAL CLR HB OUTPUT
270 008C 6A          RBIT      2,[B]      ; OPTIONAL CLR LB OUTPUT
271 008D 8E          RET          ; RETURN FROM SUBROUTINE
272 ;
273 ;
274 ;

```

TL/DD/10740-6

```

275             .FORM
276             ;
277             ; SECOND DTMF SUBROUTINE (DTMFLP) PRODUCES SIX
278             ; COMBINED LOW BAND AND HIGH BAND FREQUENCY
279             ; SINE WAVE OUTPUTS ON PINS L0 - L5
280             ;
281             ; SIX L PORT OUTPUTS (L0 - L5) FEED INTO AN EXTERNAL
282             ; RESISTOR LADDER NETWORK TO CREATE THE DTMF OUTPUT.
283             ;
284             ; FOUR VALUES FROM A KEYBOARD ROM TABLE ARE LOADED
285             ; INTO LFTBSZ (LOW FREQ TABLE SIZE), LFTADR (LOW
286             ; FREQ TABLE ADDRESS), HFTBSZ (HIGH FREQ TABLE SIZE),
287             ; AND HFTADR (HIGH FREQ TABLE ADDRESS).
288             ;
289             ; LUP42 USES THE LFPTR (LOW FREQ POINTER) AND HFPTR
290             ; (HIGH FREQ POINTER) TO ACCESS THE SINE DATA TABLES
291             ; FOR THE SELECTED FREQUENCIES ONCE PER LOOP. THESE
292             ; POINTERS ARE BOTH INCREMENTED ONCE PER LUP42.
293             ;
294             ; LUP42 PROGRAM LOOP UPDATES THE OUTPUT VALUE EVERY
295             ; 117 1/3 uSEC BY SELECTING AND THEN COMBINING NEW
296             ; VALUES FROM THE SELECTED LOW BAND AND HIGH BAND
297             ; FREQUENCY ROM TABLES WHICH SIMULATE THE SINE WAVES
298             ; FOR THE TWO FREQUENCIES.
299             ;
300             ; MULTIPLES OF THE MAGIC NUMBER OF APPROXIMATELY
301             ; 118 uSEC ARE CLOSE APPROXIMATIONS TO ALL EIGHT OF
302             ; THE DTMF FREQUENCIES.
303             ;
304             ; COP820C/840C TIMER USED TO INTERRUPT THE DTMF LUP42
305             ; PROGRAM LOOP AFTER 100 MSEC TO FINISH THE DTMF
306             ; OUTPUT AND RETURN FROM THE DTMFLP SUBROUTINE. NOTE
307             ; THAT THE STACK POINTER (SP) MUST BE ADJUSTED AFTER
308             ; THE INTERRUPT BEFORE RETURNING FROM THE SUBROUTINE.
309             ;
310             ;
311             ;
312             ;
313             ;
314             ; DECLARATIONS:
315             ;
316             0005             LFPTR   = 05             ; LOW FREQ POINTER
317             0006             TEMP    = 06             ; TEMPORARY
318             0007             HFPTR   = 07             ; HIGH FREQ POINTER
319             0008             LFTBSZ  = 08             ; LO FREQ TABLE SIZE
320             0009             LFTADR  = 09             ; LO FREQ TABLE ADDR
321             000A             HFTBSZ  = 0A             ; HI FREQ TABLE SIZE
322             000B             HFTADR  = 0B             ; HI FREQ TABLE ADDR
323             ;
324             0004             TRUN    = 04
325             ;

```

TL/DD/10740-7

```

326
327 008E BCD1FF      ;
DTMFLP: LD          PORTLC,#OFF ; INITIALIZE PORT L
328 0091 BCD01D      LD          PORTLD,#29 ; FOR NO TONE OUT
329 0094 BC0500      LD          LFPTR,#0 ; INITIALIZE OFFSET
330 0097 58          LD          B,#HFPTR ; POINTERS FOR
331 0098 9A00        LD          [B+],#0 ; DTMF SINE WAVE
332 009A A0          RC          ; TABLE LOOKUP
333 009B 65          SWAP         A ; QUADRUPLE KEY
334 009C B0          RRC          A ; VALUE AND ADD
335 009D B0          RRC          A ; OFFSET FOR KEY
336 009E 94B8        ADD          A,#0B8 ; TABLE LOOKUP
337 00A0 A6          FRLUP: X          A,[B] ; LOAD FOUR VALUES
338 00A1 AE          LD          A,[B] ; FROM ROM KEY
339 00A2 A4          LAID         ; TABLE INTO LOW
340 00A3 A2          X          A,[B+] ; FREQ LFTBSZ,
341 00A4 8A          INC          A ; LFTADR, AND HI
342 00A5 4C          IFBNE        #0C ; FREQ HFTBSZ,
343 00A6 F9          JP          FRLUP ; HFTADR
344 00A7 DEEA        LD          B,#TMRLO ; INITIALIZE TIMER
345 00A9 9A00        LD          [B+],#0 ; WITH A tC COUNT
346 00AB 9A8C        LD          [B+],#140 ; EQUIVALENT TO
347 00AD 9A00        LD          [B+],#0 ; 100 MSEC PLUS
348 00AF 9A8C        LD          [B+],#140 ; A LUP42 TIME
349 00B1 9A80        LD          [B+],#080 ; TIMER PWM, NO OUT
350 00B3 9B11        LD          [B-],#011 ; ENABLE TMR INTRPT
351 00B5 7C          SBIT         TRUN,[B] ; START TIMER
352 00B6 210F        JMP          LUP42
353
354
355
356
357 ; TELEPHONE KEY TABLE:
358
359 ; TABLE FORMAT:
360 ; PARAMETER 1: # OF LOW FREQ TABLE VALUES
361 ; PARAMETER 2: BASE ADDR. OF LOW FREQ VALUES
362 ; PARAMETER 3: # OF HIGH FREQ TABLE VALUES
363 ; PARAMETER 4: BASE ADDR. OF HIGH FREQ VALUES
364
365 ; KEY 1
366 00B8 31          .BYTE         49,02D,7,07C
367 00B9 2D
368 00BA 07
369 00BB 7C
367
368 ; KEY 2
369 00BC 31          .BYTE         49,02D,19,083
370 00BD 2D
371 00BE 13
372 00BF 83
370

```

TL/DD/10740-8


```

371                                     ; KEY 3
372 00C0 31                           .BYTE   49,02D,23,096
    00C1 2D
    00C2 17
    00C3 96
373                                     ;
374                                     ; KEY A
375 00C4 31                           .BYTE   49,02D,21,0AD
    00C5 2D
    00C6 15
    00C7 AD
376                                     ;
377                                     ; KEY 4
378 00C8 0B                           .BYTE   11,05E,7,07C
    00C9 5E
    00CA 07
    00CB 7C
379                                     ;
380                                     ; KEY 5
381 00CC 0B                           .BYTE   11,05E,19,083
    00CD 5E
    00CE 13
    00CF 83
382                                     ;
383                                     ; KEY 6
384 00D0 0B                           .BYTE   11,05E,23,096
    00D1 5E
    00D2 17
    00D3 96
385                                     ;
386                                     ; KEY B
387 00D4 0B                           .BYTE   11,05E,21,0AD
    00D5 5E
    00D6 15
    00D7 AD
388                                     ;
389                                     ; KEY 7
390 00D8 0A                           .BYTE   10,069,7,07C
    00D9 69
    00DA 07
    00DB 7C
391                                     ;
392                                     ; KEY 8
393 00DC 0A                           .BYTE   10,069,19,083
    00DD 69
    00DE 13
    00DF 83
394                                     ;
395                                     ; KEY 9
396 00E0 0A                           .BYTE   10,069,23,096
    00E1 69

```

TL/DD/10740-9

```

00E2 17
00E3 96
397
398
399 00E4 0A      ; KEY C      .BYTE      10,069,21,0AD
      00E5 69
      00E6 15
      00E7 AD
400
401
402 00E8 09      ; KEY *      .BYTE      9,073,7,083
      00E9 73
      00EA 07
      00EB 83
403
404
405 00EC 09      ; KEY 0      .BYTE      9,073,19,07C
      00ED 73
      00EE 13
      00EF 7C
406
407
408 00F0 09      ; KEY #      .BYTE      9,073,23,096
      00F1 73
      00F2 17
      00F3 96
409
410
411 00F4 09      ; KEY D      .BYTE      9,073,21,0AD
      00F5 73
      00F6 15
      00F7 AD
412
413
414
415
416      00FF      ;           . =00FF
417
418 00FF BCD01D   ; INTRPT: LD      PORTLD,#29 ; BASE LINE VALUE
419 0102 DEEF     LD      B,#PSW      ; 100 MSEC INTERRUPT
420 0104 9B00     LD      [B-],#0    ; FROM TIMER
421 0106 9E00     LD      [B],#0     ; CLR PSW AND CNTRL
422 0108 DEFD     LD      B,#SP      ; RESTORE STACK
423 010A AE       LD      A,[B]      ; POINTER (SP)
424 010B 8A       INC     A           ; TO ITS VALUE
425 010C 8A       INC     A           ; BEFORE THE
426 010D A6       X       A,[B]      ; INTERRUPT
427 010E 8E       RET                      ; RETURN FROM
428                                     ; SUBROUTINE
429
430

```

TL/DD/10740-10

```

431                                     .FORM
432                                     ;
433                                     ; LUP42 CONSISTS OF 42 COP840C INSTRUCTION CYCLE TIMES
434                                     ; LUP42 TIMING LOOP IS 42 / 0.3579545 = 117 1/3 uSEC
435                                     ;
436                                     ;
437 010F 5A      LUP42:  LD      B,#LFPTR
438 0110 AE      LD      A,[B]          ; INCREMENT LOW FREQ
439 0111 8A      INC     A              ; OFFSET POINTER
440 0112 57      LD      B,#LFTBSZ     ; TEST IF LFPTR
441 0113 82      IFEQ    A,[B]        ; BEYOND LIMIT
442 0114 64      CLR     A              ; REINITIALIZE LFPTR
443 0115 5A      LD      B,#LFPTR     ; FOR NEXT TIME
444 0116 A6      X       A,[B]
445 0117 56      LD      B,#LFTADR     ; ADD PTR TO LO FREQ
446 0118 84      ADD     A,[B]        ; TABLE ADDRESS
447 0119 A4      LAID   A              ; LOW FREQ COMPONENT
448 011A 59      LD      B,#TEMP      ; RESULT TO TEMP
449 011B A2      X       A,[B+]
450 011C AE      LD      A,[B]        ; INCREMENT HI FREQ
451 011D 8A      INC     A              ; OFFSET POINTER
452 011E 55      LD      B,#HFTBSZ    ; TEST IF HFPTR
453 011F 82      IFEQ    A,[B]        ; BEYOND LIMIT
454 0120 64      CLR     A              ; REINITIALIZE HFPTR
455 0121 58      LD      B,#HFPTR     ; FOR NEXT TIME
456 0122 A6      X       A,[B]
457 0123 54      LD      B,#HFTADR     ; ADD PTR TO HI FREQ
458 0124 84      ADD     A,[B]        ; TABLE ADDRESS
459 0125 A4      LAID   A              ; HI FREQ COMPONENT
460 0126 59      LD      B,#TEMP      ; ADD LOW FREQ VALUE
461 0127 84      ADD     A,[B]        ; TO HI FREQ VALUE
462 0128 9CDO    X       A,PORTLD     ; RESULT TO PORT L
463 012A A4      LAID   A              ; EQUIVALENT OF
464 012B A4      LAID   A              ; SIX NOP'S
465 012C E2      JP      LUP42        ; TIMING LOOP OF
466                                     ; 117 1/3 uSEC
467                                     ;
468                                     ;
469                                     ;
470                                     ;

```

TL/DD/10740-11

```

471             .FORM
472             ;
473             ; THE FREQUENCY APPROXIMATION IS EQUAL TO THE NUMBER OF
474             ; CYCLES OF SINE WAVE DIVIDED BY THE TIME IN THE TOTAL
475             ; NUMBER OF LUP42 CYCLES BEFORE THE REPETITION OF THE
476             ; ROM TABLE. AS AN EXAMPLE, CONSIDER THE THREE CYCLES
477             ; OF SINE WAVE AND 19 VALUES IN THE ASSOCIATED 1336 HZ
478             ; ROM TABLE. THE 19 CYCLES OF LUP42 TIMES THE LUP42
479             ; TIME OF 117 1/3 uSEC IS DIVIDED INTO THE THREE CYCLES
480             ; OF SINE WAVE TO YIELD A VALUE OF 1345.69 HZ AS THE
481             ; 1336 HZ APPROXIMATION.
482             ;
483             ; THE VALUES IN THE ROM TABLES FOR THE DTMF SINE WAVES
484             ; SHOULD WRAP AROUND END TO END IN EITHER DIRECTION TO
485             ; FORM A SYMETRICAL LOOP. THE FIRST VALUE IN THE ROM
486             ; TABLE REPRESENTS THE BASE LINE FOR THAT FREQUENCY.
487             ;
488             ; THE HIGH BAND DTMF FREQUENCIES HAVE A BASE LINE VALUE
489             ; OF 16 AND A MAXIMUM VALUE OF 32. THE LOW BAND DTMF
490             ; FREQUENCIES HAVE A BASE LINE VALUE OF 13 AND A
491             ; MAXIMUM VALUE OF 26. THIS DIFFERENCE IN BASE LINE
492             ; VALUES IS NECESSARY TO SATISFY THE REQUIREMENT OF THE
493             ; HIGH BAND FREQUENCIES NEEDING A LEVEL 2 dB ABOVE THE
494             ; LEVEL OF THE LOW BAND FREQUENCIES TO COMPENSATE FOR
495             ; LOSSES IN TRANSMISSION. THE SUM OF THE TWO BASE LINE
496             ; VALUES YIELDS A BASE LINE VALUE OF 29, WHILE THE SUM
497             ; OF THE TWO MAXIMUM VALUES YIELDS A MAXIMUM VALUE OF
498             ; 58. THUS THE SIX BIT DTMF OUTPUT FROM THE L PORT TO
499             ; THE LADDER NETWORK RANGES FROM 0 TO 58, WITH A BASE
500             ; LINE VALUE OF 29.
501             ;
502             ; THE VALUES IN THE DTMF SINE WAVE TABLES ARE
503             ; CALCULATED BY COMPUTING THE SINE VALUE AT THE
504             ; APPROPRIATE POINTS, SCALING THE SINE VALUE UP TO THE
505             ; BASE LINE VALUE, AND THEN ADDING THE RESULT TO THE
506             ; BASE LINE VALUE. THE FOLLOWING EXAMPLE WILL HELP TO
507             ; CLARIFY THIS CALCULATION.
508             ;
509             ; CONSIDER THE THREE CYCLES OF SINE WAVE ACROSS 19
510             ; DATA POINTS FOR THE 1336 HZ DTMF HIGH BAND FREQUENCY.
511             ; THE FIRST VALUE IN THE TABLE IS THE BASE LINE VALUE
512             ; OF 16. WITH 2 PI RADIANS PER SINE WAVE CYCLE,
513             ; THE SUCCEEDING VALUES IN THE TABLE REPRESENT THE
514             ; SINE VALUES OF 1 X (6 PI / 19), 2 X (6 PI / 19),
515             ; 3 X (6 PI / 19), . . . . , UP TO 18 X (6 PI / 19).
516             ; LET US NOW CONSIDER THE SEVENTH AND EIGHTH VALUES
517             ; IN THE TABLE, REPRESENTING THE SINE VALUES OF
518             ; 6 X (6 PI / 19) AND 7 X (6 PI / 19) RESPECTIVELY.
519             ; THE CALCULATIONS OF 16 X SIN [6 X (6 PI / 19)] AND
520             ; 16 X SIN [7 X (6 PI / 19)] YIELD VALUES OF - 5.20 AND
521             ; 9.83 RESPECTIVELY. ROUNDED TO THE NEAREST INTEGER

```

TL/DD/10740-12

```

522 ; GIVES VALUES OF - 5 AND 10. WHEN ADDED TO THE BASE
523 ; LINE VALUE OF 16, THESE VALUES YIELD THE RESULTS
524 ; 11 AND 26 FOR THE SEVENTH AND EIGHTH VALUES IN THE
525 ; 1336 HZ DTMF TABLE. SYMMETRY IN THE LOOP OF 19 VALUES
526 ; IN THE DTMF TABLE DICTATES THAT THE FOURTEENTH AND
527 ; THIRTEENTH VALUES IN THE TABLE ARE 21 AND 6,
528 ; REPRESENTING VALUES OF 5 AND - 10 FROM THE
529 ; CALCULATIONS.
530 ;
531 ; THE AREA UNDER A HALF CYCLE OF SINE WAVE RELATIVE TO
532 ; THE AREA OF THE SURROUNDING RECTANGLE IS 2/PI, WHERE
533 ; PI RADIANS REPRESENT THE SINE WAVE HALF CYCLE. THIS
534 ; SURROUNDING RECTANGLE HAS A LENGTH OF PI AND A HEIGHT
535 ; OF 1, WITH THE HEIGHT REPRESENTING THE MAXIMUM SINE
536 ; VALUE. CONSEQUENTLY, THE AREA OF THIS SURROUNDING
537 ; RECTANGLE IS PI. THE INTEGRAL OF THE AREA UNDER THE
538 ; HALF SINE WAVE FROM 0 TO PI IS EQUAL TO 2. THE RATIO
539 ; OF 2/PI IS EQUAL TO 63.66 % , SO THAT THE TOTAL OF
540 ; THE VALUES FOR EACH HALF SINE WAVE SHOULD APPROXIMATE
541 ; 63.66 % OF THE SUM OF THE MAX VALUES. THE MAXIMUM
542 ; VALUES (RELATIVE TO THE BASE LINE) ARE 13 AND 16
543 ; RESPECTIVELY, FOR THE LOW AND HIGH BAND FREQUENCIES.
544 ;
545 ;
546 ;
547 ;
548 ;
549 ; LF697: 4 CYCLES OF SINE WAVE SPREAD
550 ; ACROSS 49 TIMING LOOP (LUP42) CYCLES
551 ;
552 ; FREQ. = 4 / (49 X 117 1/3) = 695.73 HZ
553 ; ERROR = (697 - 695.73) / 697 = - 0.18 %
554 ;
555 012D 0D .BYTE 13,19,24,26,25,20,14,7,2,0
012E 13
012F 18
0130 1A
0131 19
0132 14
0133 0E
0134 07
0135 02
0136 00
556 0137 01 .BYTE 1,5,11,18,23,26,25,21,15,9
0138 05
0139 0B
013A 12
013B 17
013C 1A
013D 19
013E 15

```

TL/DD/10740-13

```

013F 0F
0140 09
557 0141 03          .BYTE      3,0,1,4,10,16,22,25,26,23
0142 00
0143 01
0144 04
0145 0A
0146 10
0147 16
0148 19
0149 1A
014A 17
558 014B 11          .BYTE      17,11,5,1,0,3,8,15,21,25
014C 0B
014D 05
014E 01
014F 00
0150 03
0151 08
0152 0F
0153 15
0154 19
559 0155 1A          .BYTE      26,24,19,12,6,1,0,2,7
0156 18
0157 13
0158 0C
0159 06
015A 01
015B 00
015C 02
015D 07
560          ;
561          ;
562          ; LF770:   1 CYCLE OF SINE WAVE SPREAD
563          ;                   ACROSS 11 TIMING LOOP (LUP42) CYCLES
564          ;
565          ;                   FREQ. = 1 / (11 X 117 1/3) = 774.79 HZ
566          ;                   ERROR = (774.79 - 770) / 770 = + 0.62 %
567          ;
568 015E 0D          .BYTE      13,20,25,26,23,17,9,3,0,1
015F 14
0160 19
0161 1A
0162 17
0163 11
0164 09
0165 03
0166 00
0167 01
569 0168 06          .BYTE      6
570          ;

```

TL/DD/10740-14

```

571 ;
572 ; LF852: 1 CYCLE OF SINE WAVE SPREAD
573 ; ACROSS 10 TIMING LOOP (LUP42) CYCLES
574 ;
575 ; FREQ. = 1 / (10 X 117 1/3) = 852.27 HZ
576 ; ERROR = (852.27 - 852) / 852 = + 0.03 %
577 ;
578 0169 0D .BYTE 13,21,25,26,21,13,5,1,0,5
    016A 15
    016B 19
    016C 1A
    016D 15
    016E 0D
    016F 05
    0170 01
    0171 00
    0172 05
579 ;
580 ;
581 ; LF941: 1 CYCLE OF SINE WAVE SPREAD
582 ; ACROSS 9 TIMING LOOP (LUP42) CYCLES
583 ;
584 ; FREQ. = 1 / (9 X 117 1/3) = 946.97 HZ
585 ; ERROR = (946.97 - 941) / 941 = + 0.63 %
586 ;
587 0173 0D .BYTE 13,21,26,24,18,8,2,0,5
    0174 15
    0175 1A
    0176 18
    0177 12
    0178 08
    0179 02
    017A 00
    017B 05
588 ;
589 ;
590 ;
591 ; HF1209: 1 CYCLE OF SINE WAVE SPREAD
592 ; ACROSS 7 TIMING LOOP (LUP42) CYCLES
593 ;
594 ; FREQ. = 1 / (7 X 117 1/3) = 1217.53 HZ
595 ; ERROR = (1217.53 - 1209) / 1209 = + 0.71 %
596 ;
597 017C 10 .BYTE 16,29,32,23,9,0,3
    017D 1D
    017E 20
    017F 17
    0180 09
    0181 00
    0182 03
598 ;

```

TL/DD/10740-15

```

599          ;
600          ; HF1336:  3 CYCLES OF SINE WAVE SPREAD
601          ;                ACROSS 19 TIMING LOOP (LUP42) CYCLES
602          ;
603          ;                FREQ. = 3 / (19 X 117 1/3) = 1345.69 HZ
604          ;                ERROR = (1345.69 - 1336) / 1336 = + 0.73 %
605          ;
606 0183 10          .BYTE      16,29,31,19,4,0,11,26,32,24
      0184 1D
      0185 1F
      0186 13
      0187 04
      0188 00
      0189 0B
      018A 1A
      018B 20
      018C 18
607 018D 08          .BYTE      8,0,6,21,32,28,13,1,3
      018E 00
      018F 06
      0190 15
      0191 20
      0192 1C
      0193 0D
      0194 01
      0195 03
608          ;
609          ;
610          ; HF1477:  4 CYCLES OF SINE WAVE SPREAD
611          ;                ACROSS 23 TIMING LOOP (LUP42) CYCLES
612          ;
613          ;                FREQ. = 4 / (23 X 117 1/3) = 1482.21 HZ
614          ;                ERROR = (1482.21 - 1477) / 1477 = + 0.35 %
615          ;
616 0196 10          .BYTE      16,30,29,14,1,4,20,32,26,10
      0197 1E
      0198 1D
      0199 0E
      019A 01
      019B 04
      019C 14
      019D 20
      019E 1A
      019F 0A
617 01A0 00          .BYTE      0,8,24,32,22,6,0,12,28,31
      01A1 08
      01A2 18
      01A3 20
      01A4 16
      01A5 06
      01A6 00

```

TL/DD/10740-16


```

01A7 0C
01A8 1C
01A9 1F
618 01AA 12          .BYTE    18,3,2
01AB 03
01AC 02

619          ;
620          ;
621          ; HF1633:  4 CYCLES OF SINE WAVE SPREAD
622          ;                ACROSS 21 TIMING LOOP (LUP42) CYCLES
623          ;
624          ;                FREQ. = 4 / (21 X 117 1/3) = 1623.38 HZ
625          ;                ERROR = (1633 - 1623.38) / 1633 = - 0.59 %
626          ;
627 01AD 10          .BYTE    16,31,27,9,0,11,29,30,14,0
01AE 1F
01AF 1B
01B0 09
01B1 00
01B2 0B
01B3 1D
01B4 1E
01B5 0E
01B6 00
628 01B7 07          .BYTE    7,25,32,18,2,3,21,32,23,5
01B8 19
01B9 20
01BA 12
01BB 02
01BC 03
01BD 15
01BE 20
01BF 17
01C0 05
629 01C1 01          .BYTE    1
630          ;
631          ;
632          ;

```

TL/DD/10740-17

```

633          .FORM
634          ;
635          ; DTMF KEYBOARD DECODE SUBROUTINE (KBRDEC)
636          ;
637          ; KEYBOARD INPUT DATA IS IN ACCUMULATOR WITH A
638          ; LOW TRUE FORMAT AS FOLLOWS:
639          ;     BITS 7 TO 4 : LOW TRUE COLUMN VALUE (E,D,B,7)
640          ;     BITS 3 TO 0 : LOW TRUE ROW VALUE (E,D,B,7)
641          ;
642          ; ASSUMPTION MADE THAT COLUMN STROBES (LOW TRUE) ARE
643          ; OUTPUT, WHILE ROW VALUES (LOW TRUE) ARE INPUT.
644          ;
645          ; LOW TRUE COLUMN/ROW INPUT DIGIT IN ACCUMULATOR IS
646          ; TRANSFORMED INTO A DTMF HEX DIGIT KEY VALUE
647          ;
648          ; TABLE LOOKUP TRANSFORMATION CHECKS FOR MULTIPLE KEYS,
649          ; NO KEY, OR NO COLUMN SELECT, AND THEN PRODUCES
650          ; A DTMF HEX DIGIT KEY VALUE WITH A BINARY FORMAT
651          ; OF 0000RCC FOR A SINGLE KEY INPUT,
652          ; WHERE - RR IS LOW BAND (LB) FREQUENCY SELECT
653          ;       - CC IS HIGH BAND (HB) FREQUENCY SELECT
654          ;
655          ; KBRDEC SUBROUTINE IS EXITED WITH A RETURN (RET)
656          ; COMMAND TO INDICATE MULTIPLE KEYS, NO KEY,
657          ; OR NO COLUMN SELECT
658          ;
659          ; KBRDEC SUBROUTINE IS EXITED WITH A RETURN AND SKIP
660          ; (RETSK) COMMAND TO INDICATE A SINGLE KEY ENTRY
661          ;
662          ;
663          0200          . =0200
664          ;
665          ; LOW TRUE TRANSLATION TABLE - ONLY E,D,B,7 ACCEPTABLE
666          ;
667          0200 C0          .BYTE 0C0,0C0,0C0,0C0,0C0,0C0,0C0,0C
        0201 C0
        0202 C0
        0203 C0
        0204 C0
        0205 C0
        0206 C0
        0207 0C
668          0208 C0          .BYTE 0C0,0C0,0C0,8,0C0,4,0,0C0
        0209 C0
        020A C0
        020B 08
        020C C0
        020D 04
        020E 00
        020F C0
669          ;

```

TL/DD/10740-18

```

670          ;
671 0210 5F      KBRDEC: LD      B,#KDATA
672 0211 A6      X          A,[B] ; STORE LOW TRUE
673 0212 AE      LD          A,[B] ; COLUMN/ROW VALUE
674 0213 95F0    AND        A,#0F0 ; EXTRACT LOW TRUE COLUMN
675 0215 65      SWAP       A          ; & PUT IN LOWER NIBBLE
676 0216 A4      LAID       ; 0000CC00 FROM TABLE
677 0217 A0      RC          ; SHIFT TABLE VALUE DOWN
678 0218 B0      RRC        A          ; TWO BITS TO PRODUCE
679 0219 B0      RRC        A          ; 000000CC
680 021A A6      X          A,[B] ; STORE RESULT
681 021B 950F    AND        A,#0F ; EXTRACT LOW TRUE ROW
682 021D A4      LAID       ; 0000RR00 FROM TABLE
683 021E 84      ADD        A,[B] ; ADD TO PRODUCE 0000RRCC
684 021F 930F    IFGT       A,#0F ; RETURN IF MULTIPLE KEYS,
685 0221 8E      RET         ; NO KEYS, OR NO COLUMN
686 0222 8D      RETSK      ; RETURN AND SKIP
687          ;          ; IF SINGLE KEY
688          ;
689          ;
690          ;
691          .END

```

TL/DD/10740-19

B	00FE	BYP1	0072	BYP2	0075	BYPA	0019
BYPB	001B	CNTRL	00EE	DTMFGP	0040	DTMFLP	008E
FINI	0086	FRLUP	00A0	HFPTR	0007	HFTADR	000B
HFTBSZ	000A	INTRPT	00FF *	KBRDEC	0210	KDATA	0000
LFPTR	0005	LFTADR	0009	LFTBSZ	0008	LOOP	0006
LUP	004D	LUP1	006C	LUP2	0078	LUP42	010F
PORTD	00DC	PORTGC	00D5	PORTGD	00D4	PORTI	00D7
PORTLC	00D1	PORTLD	00D0	PSW	00EF	RO	00F0
R1	00F1	R2	00F2	R3	00F3	SP	00FD
START	0000 *	TAUHI	00ED *	TAULO	00EC *	TEMP	0006
TMRHI	00EB *	TMRLO	00EA	TRUN	0004	X	00FC

TL/DD/10740-20

LIFE SUPPORT POLICY

NATIONAL'S PRODUCTS ARE NOT AUTHORIZED FOR USE AS CRITICAL COMPONENTS IN LIFE SUPPORT DEVICES OR SYSTEMS WITHOUT THE EXPRESS WRITTEN APPROVAL OF THE PRESIDENT OF NATIONAL SEMICONDUCTOR CORPORATION. As used herein:

1. Life support devices or systems are devices or systems which, (a) are intended for surgical implant into the body, or (b) support or sustain life, and whose failure to perform, when properly used in accordance with instructions for use provided in the labeling, can be reasonably expected to result in a significant injury to the user.
2. A critical component is any component of a life support device or system whose failure to perform can be reasonably expected to cause the failure of the life support device or system, or to affect its safety or effectiveness.



National Semiconductor Corporation
 2900 Semiconductor Drive
 P.O. Box 58090
 Santa Clara, CA 95052-8090
 Tel: 1(800) 272-9959
 TWX: (910) 339-9240

National Semiconductor GmbH
 Livny-Gargan-Str. 10
 D-82256 Fürstenfeldbruck
 Germany
 Tel: (81-41) 35-0
 Telex: 527849
 Fax: (81-41) 35-1

National Semiconductor Japan Ltd.
 Sumitomo Chemical
 Engineering Center
 Bldg. 7F
 1-7-1, Nakase, Mihama-Ku
 Chiba-City,
 Chiba Prefecture 261
 Tel: (043) 299-2300
 Fax: (043) 299-2500

National Semiconductor Hong Kong Ltd.
 13th Floor, Straight Block,
 Ocean Centre, 5 Canton Rd.
 Tsimshatsui, Kowloon
 Hong Kong
 Tel: (852) 2737-1600
 Fax: (852) 2736-9960

National Semicondutores Do Brazil Ltda.
 Rue Deputado Lacorda Franco
 120-3A
 Sao Paulo-SP
 Brazil 05418-000
 Tel: (55-11) 212-5066
 Telex: 391-1131931 NSBR BR
 Fax: (55-11) 212-1181

National Semiconductor (Australia) Pty, Ltd.
 Building 16
 Business Park Drive
 Monash Business Park
 Nottingham, Melbourne
 Victoria 3168 Australia
 Tel: (3) 558-9999
 Fax: (3) 558-9998

National does not assume any responsibility for use of any circuitry described, no circuit patent licenses are implied and National reserves the right at any time without notice to change said circuitry and specifications.